
Controlling the 

WORLDWORLD  
from your Armchair. 

 

Sony, Infrared Remote Control Decoding. 
 

Infrared remote control decoding is considered something of a black art, 
however, this tutorial will show you that its principals are quite straight-
forward, and easy to implement on a PIC microcontroler. 
 
Infrared remote control has been around for a very long time now, and we tend to take it for granted. Yet, 
it is a marvel of modern technology, which allows a whole variety of devices to be activated with the 
touch of a button. Remote control handsets are so abundant that they may be purchased new for a few 
pounds, which makes them viable items for experimentation. There are dedicated chips available that will 
decode the signals from a particular handset, however, with the flexibility and cost effectiveness of the 
PIC range of microcontrollers we can develop a decoding subroutine that may be placed into your own 
programs, or used as a standalone infrared to RS232 converter. 
 
Manufacturers Protocols. 
Regrettably, remotes do not come in a single flavour, each manufacturer uses a different set of protocols. 
The three main ones are RC80, which is used by Panasonic. RC5, which was designed by Philips and is 
one of the more popular types, and then there’s the Sony protocol, named S.I.R.C, which is hugely popu-
lar and also one of the simplest to decode. Therefore, I will take the less complex type and endeavour to 
illustrate how to decode the signals from a Sony remote control handset, using the ever-popular 
PIC16F84 microcontroller. 
 
Infrared to TTL Converter. 
In a bid to eliminate ambient light sources, both natural and manmade, from interfering with the data 
stream transmitted by the handset, modulated light is used. This modulation is centred around different 
frequencies depending on the manufacturer; and varies from 32kHz to 40kHz. In the case of Sony hand-
sets, the modulation is centred at 40kHz, which means we require a device that can receive the modulated 
infrared light and convert it into a TTL signal that the PIC can handle. 
 
There are a number of these devices available, each having a specific centre frequency that they’re more 
sensitive too.  The device used for this tutorial is the IS1U60 from Sharp. It has a centre frequency of 
38Khz, which is close enough to 40kHz so as not to matter. Figure 1, shows the internal block diagram of 
one of these devices. 
 
 
 
 
 

Fig 1. Internals of the IS1U60. 
 
As you can see, these deceptively simple looking devices are a lot more than a re-packaged IR Photodi-
ode. They filter, amplify and demodulate the infrared signal. Then give a nice clean TTL output  by 
means of a final comparator stage. They also have a built in automatic gain control (AGC), which helps  

stop overloading; if the handset is held too close. Using one of these devices is a great deal cheaper (and 
easier) than building your own discrete version.  
Most IR sensors have an active low output, which means that the PIC is presented with a logic 0 when an 
infrared signal is detected. With no signal present, a maximum current of 4.8mA is consumed (2.8mA 
being typical). In addition, the recommended voltage is 4.7V to 5.3V. 
 
Sony Protocol (S.I.R.C). 
SIRC (Serial Infra-Red Control) uses a form of pulse width modulation (PWM) to build up a 12-bit serial 
interface, known as a packet. This is the most common protocol, but 15-bit and 20-bit versions are also 
available. A pulse with a duration of 2.4ms is sent first as a header, this allows the internal AGC to adjust 
and also allows the receiver to check if a valid packet is being received. A 1-bit is represented by a pulse 
duration of 1.2ms, while a 0-bit has a duration of 0.6ms. A delay of 0.6ms is placed between every pulse.  
 
The string of pulses build up the 12-bit packet consisting of a 5-bit (0..31) device code, which represents a 
TV, Video, Hi-Fi etc (see table 1), and a 7-bit (0..127) button code, which represents the actual button 
pressed on the remote (see table 2). The packet is transmitted most significant bit first (MSB), with the 
device code being sent, then the button code. Figure 2, illustrates this more clearly. After the packet is 
sent, a delay is implemented, which brings the whole transmitted signal to a length of 45ms. This is re-
peated for as long as a button is pressed. 
 
 
 
 
 
 
 
 

Fig 2. 12-bit packet construction. 
 

 
 
 
 
 
 
 
 
 
 

 
Assembler vs BASIC. 
Knowing the principals behind infrared communications is one 
thing, actually writing software based on the information is a 
whole new ball game. Whenever PIC micros are mentioned, peo-
ple tend to think of the rather cryptic language of assembler, how-
ever, this is not the case anymore, as there are many high level 
language implementations for use with the PIC, such as C, Pascal, 
and BASIC. My personal preference is BASIC, or rather 
PicBASIC Pro. The BASIC language in general has received a lot of bad press since its conception in the 
middle part of the 70’s and is considered to be clumsy and inflexable, yet nothing could be further from 
the truth. Thanks to microEngineering lab’s PicBASIC and PicBASIC Pro compilers, and Leading Edge 
Technologies PicBASIC compiler range, this language has been brought into the 21st century.  

Command Function 
0-9 Numerals 0 to 9 
16 Channel + 
17 Channel - 
18 Volume + 
19 Volume - 
20 Mute 
21 Power 
22 Reset 
23 Audio mode 
24 Contrast + 
25 Contrast - 
26 Colour + 
27 Colour - 
30 Brightness + 
31 Brightness - 
38 Balance left 
39 Balance right 
47 Power off 

Table 2. SIRC TV button code. 

Command  Device 
1 Television  receiver 
2 VCR 1 
4 VCR 2 
6 Laser disk player 

12 Surround sound unit 
16 Cassette deck/tuner 
17 CD player 
18 Equaliser 

Table 1. SIRC device code. 

 Fig 3. Sony, Infrared Remote control decoder circuit. 

 
Thanks also, in part, to BASIC’s shallow learning curve, software designs that used to take weeks can 
now be realised in a just few hours.  
 
So as to not seem too biased towards either language, I will present the software for this article in both 
assembler and PicBASIC Pro, which will enable you to choose your preferred type. I will also endeavor 
to illustrate the pro’s and cons (no pun intended) of both languages by not using optimised assembler rou-
tines. This means that both the BASIC and the assembler versions will follow the same structuring, which 
will enable a fairer appraisal of them. 
 
It is not my intention to teach you how to program a PICmicro, therefore, throughout this article, it is as-
sumed that you already have some knowledge of either assembler or PicBASIC Pro. And that you have a 
means of programming the PIC16F84. For more information concerning the PicBASIC range of compil-
ers, as well as an assortment of programmers, visit Crownhill Associate’s dedicated web site at 
www.picbasic.co.uk. For information concerning assembler programming, visit Microchip’s web site at 
www.microchip.com.  
 
Circuit Description. 
In order to demonstrate the 
principals behind infrared de-
coding, the circuit in figure 3 is 
employed. The PIC circuit in-
corporates two light emitting 
diodes, one green and the other, 
red. The software is arranged in 
such a way that by pressing the 
channel-up button on a TV re-
mote, the green LED will illu-
minate and channel-down will 
illuminate the red LED.  
 
As well as illuminating the 
LED’s, two bytes are transmit-
ted serially (Async RS232) from 
PortA,3 through a 1kΩ current limiting resistor (R2). The serial data contains the device code as well as 
the button code and is transmitted at inverted 9600 baud (N-8-1). Possible uses for this could be to attach 
it to the PC’s serial input for remotely controlling some software, or to attach it to a BASIC Stamp for use 
in a robotics construction.  
 
The circuit layout is not too critical and could easily be built on a piece of stripboard. However, decoup-
ling capacitor C5 should be placed as close to the IR sensor as possible, and C2 should also be located 
close to the PIC. 
 
Getting down to the coding. 
The actual infrared decoding software is presented in the form of a subroutine, named IRIN, which will 
ease the inclusion of it into your own programs. The subroutine and subsequent main program loop may 
be split into several software tasks. These are outlined below: - 
 
Task 1… Configure PortA as both Inputs and Outputs 
 
Task 2… Devise a method of measuring the high to low pulse length received from the active low IR 
sensor. 

Bsf STATUS,5    ;Point to TRIS reg 
Movlw b’00000100’ ;Set PortA,2 as IN 
Movwf PortA    ;Configure the Port 
Bcf STATUS,5    ;Back to Page0 
 
Listing 1. Assembler, Port direction. 

 

; Measure the duration of a high to low pulse on PortA,2. 
; And leave the result in P_VAL. 
; An 11 microsecond resolution is achieved with a 4MHz crystal. 
Pulsin  Clrwdt  ; Walk the dog 
 Clrf     Cntr ; Clear the variables used, 
         Clrf     P_Val ; prior to the subroutine 
Trans  Btfss    PortA,2 ; Wait for a 1 to 0 transition 
           Goto     Edge ; Edge found! 
         Incfsz   P_Val ; Else, increment P-VAL until >255 
         Goto     Trans 
        Incfsz   Cntr ; Loop until 255 
         Goto     Trans 
 Return 
Edge   Clrf     P_Val ; A 1 to 0 transition occurred 
Ege_lp Btfsc    PortA,2 ; Count how long it's logic 0  

Return 
 Clrwdt  ; Walk the dog  
 Nop  ; Timing loop 
 Nop 
 Nop 
 Nop 
 Nop 
         Incfsz   P_Val   ; Increment P_VAL until >255 
         Goto     Edge_LP 
 Return 
 

Listing 2. Assembler, pulse measurement subroutine. 

 
Task 3… Implement task 2 to detect the header and bit pulses and then construct the 12-bit packet.  
 
Task 4… Split the packet into two separate bytes containing the 7-bit button and 5-bit device codes. 
 
Task 5… Devise a method of transmitting inverted serial RS232 data. 
 
Task 6… Construct a main program loop that calls the decoder subroutine and illuminates the correct 
LED, as well as using task 5 for transmitting both, the device and button codes, serially. 
 
Task 1. 
Our first coding task, that of configuring the Port’s direction, is the easiest to accomplish. The assembler 
code for this is shown in listing 1. This will configure bits 0, 1 and 3 of PortA as outputs, for the attach-
ment of the LEDs as well as the serial output. Bit-2 is made an input for the attachment of the IR sensor. 
 
The same thing written in PicBASIC Pro is: - 
 
TrisA = %00000100 
 
Note, that there is no actual need to do this in PicBASIC, as the 
commands that deal with external influences, automatically set the 
required pins as inputs or outputs. 
 
Task 2. 
Our second coding task, is a means of measuring the pulse durations that signify a header, as well as the 
separate ones and zeros that go to make up the packet. An assembler version of a routine that will do just 
this is shown in listing 2.  
 
The high to low pulse duration is measured at bit-2 of PortA and the 8-bit value is returned in the variable 
P_VAL. Because we’re using an 8-bit (0..255) variable, it’s impossible to return a value of 2400 for a 
pulse length of 2400 microseconds. Therefore, the routine has a resolution of approx 11microseconds 
when used in conjunction with a 4MHz crystal. 
An 11us resolution was chosen as opposed to 
10us, because not all remote handsets stick 
stringently to the recommended pulse widths. 
Therefore, a header pulse could be more than 
2.55ms in length, which would push it beyond a 
byte’s storage capacity, i.e. greater than 255. 
The values returned in P_VAL for a given 
pulse length are as follows: - 
 
Header pulse… 2400us. will return 220. 
One-bit pulse… 1200us. will return 110. 
Zero-bit pulse… 600us. will return 55. 
 
To do the same task in PicBASIC Pro, requires 
just one command: - 
 
Pulsin PortA.2 , 0 , (8 or 16-bit Variable) 
 
When used in association with a 4MHz crystal, 
the compiler’s PULSIN command has a resolu-
tion of 10 microseconds. Also, if a 16-bit vari- 

' Receive a signal from a Sony remote control, 
' If no header then IR_DEV, IR_BUT will hold 255 
IRIN Clrwdt  ; Walk the dog 
 Call Pulsin ; Measure the header 
; Verify a good header, if its not valid then exit 
; ** If PVAL < 200 then return with IR_DEV=255 ** 
 Movlw  200 
 Subwf P_VAL,W 
 Btfsc STATUS,C 
 Goto Next1 
 Movlw 255 
 Movwf IR_Dev 
 Return 
; ** If PVAL > 250 then return with IR_DEV=255 ** 
Next1 Movlw  250 
 Subwf P_VAL,W 
 Btfss STATUS,C 
 Goto PK_Strt 
 Movlw 255 
 Movwf IR_Dev 
 Return 
; Build up the packet, by pulling in all 12-bits  
PK_Strt Movlw 12 ; Create a loop for 12-bits 
 Movwf  Bitcnt 
S_again Call  Pulsin ; Get the bit duration 
 Movlw 80 ; If it’s >= 80 then it's a 1 
 Subwf P_VAL,w 
 Btfsc STATUS,C 
 Goto One 
 Bcf  Packet+1,4; Clear the bit 
 Goto  Cont 
One Bsf  Packet+1,4; Set the bit 
Cont Rrf  Packet+1,F; Rotate the bit into place 
 Rrf  Packet,F 
 Decfsz  Bitcnt ; Have we done 12-bits yet? 
 Goto  S_again ; No! then loop again 
  

Listing 3. Assembler, 12-bit Packet constuction. 

' Receive a signal from a Sony remote control, 
' and return with the 7-bit BUTTON code in the variable 
' IR_BUT and the 5-bit DEVICE code in the variable IR_DEV. 
' If header no header then IR_DEV, IR_BUT will hold 255 
IRIN: IR_Dev=255:IR_But=255 ‘ Preset the return variables 

Pulsin PortA.2,0,Header ' Measure the header length. 
         If Header < 200 then Return' Verify a good header     

If Header > 270 then Return' If not valid then exit 
' Receive the 12 data bits and convert them into a packet 
 For Bitcnt=0 to 11  ' Create a loop of 12 
 Pulsin PortA.2,0,P_Val ' Receive the IR bit pulse 
 If P_Val >= 90 then  ' If it's >= 90 then we've  

‘ received a 1 
 Packet.0[Bitcnt]=1 ' So set the appropriate bit 
 Else    ' Else 
 Packet.0[Bitcnt]=0 ' Clear the appropriate bit 
 Endif 
 Next    ' Close the loop 
 

Listing 4. PicBASIC Pro, 12-bit Packet constuction. 

 
able is used to hold the result then a duration of 0.. 65535us may be measured, where as, if an 8-bit vari-
able is used this is reduced to 0..255us . We can use this property to our advantage by detecting the 
2400us header pulse with a 16-bit variable, and the individual 600us or 1200us bit pulses with an 8-bit 
variable. This will eliminate any problems arising from a header pulse that is longer than 2.55ms. The 
values returned from the PULSIN command are as follows: - 
 
Header pulse…2400us. will return 240. 
One-bit pulse… 1200us. will return 120. 
Zero-bit pulse… 600us. will return 60. 
 
The middle parameter of the PULSIN command (0 or 1), determines whether a high to low pulse, or a 
low to high pulse is to be measured. Where a zero, measures a high to low type. 
 
Task 3. 
We now come to one; of the two main body parts that 
build up the subroutine IRIN, in which we gather the 
bit information received from the IR sensor and con-
struct the 12-bit packet.  
 
The assembler version of this is shown in listing 3. The 
first thing the routine does is to try and detect a 2.4ms 
header pulse using the PULSIN subroutine (task 1). 
The result, held in P_VAL is examined to see whether 
it’s between the values of 200 and 250. If it does not lie 
between these values, then the subroutine is exited with 
IR_DEV and IR_BUT holding a value of 255, which 
signifies an invalid header. If, however, a valid header 
IS detected, then a loop of 12 is set up. Within this 
loop, the individual bits are measured using the subrou-
tine, PULSIN. Depending on the result returned in 
P_VAL, the individual bits of the 16-bit variable 
PACKET are set or cleared. This is achieved by split-
ting the difference between a one-bit (110), and a zero-
bit (55). If the result is greater than or equal to 80 then 
it must be a one-bit that’s been received and if it’s less 
than 80 then it must be a zero-bit. The PicBASIC Pro 

version of the same routine is shown in listing 
4. This has exactly the same function as the 
assembler version, however, because of the 
different values returned from the PULSIN 
command, the comparisons for a header and bit 
pulses are slightly different. The resulting 12-
bit packet for both types of routine are held in 
the variable PACKET, ready for splitting into 
its separate codes. 

; Transmit the byte held in W at inverted 9600 baud (8-N-1)  
; From PortA,3 
Sout Movwf Tr_Byte ; Load TR_BYTE with W reg 

Movlw 08 
 Movwf Bit_Cntr ; Create a loop of 8 
 Bsf PortA,3 ; Send the start bit 
 Call Bit_Dly ; Delay one bit time   
Xmtlp Rrf Tr_Byte ; Rotate Right, moves data bits 

; into Carry, starting with bit-0.  
    Btfsc    Status,0 ; Is it a One-bit? 
 Bcf PortA,3 ; Yes, so send A One 
     Btfss    Status,0 ; Is it a Zero-bit? 
 Bsf PortA,3 ; Yes, so send A Zero 
 Call Bit_Dly ; Delay one bit time   
 Decfsz Bit_Cntr ; Have we reached 8-bits yet? 
 Goto Xmtlp ; No, so loop again 
 Bcf PortA,3 ; Yes, so send the stop bit  
 Call Bit_Dly ; Delay one bit time    

Return 
; ** Delay 1-bit time subroutine** 
Bit_Dly Movlw 22 ; Set Baud to 9600 
 Movwf Dlctr 
Slp Clrwdt  ; Walk the dog (1us) 
 Decfsz Dlctr 
 Goto Slp 
 Return 
 

Listing 7. Assembler, Serial output subroutine. 

 
Task 4. 
For the resulting 12-bit packet to be of any practical 
use, it must be split into the 5-bit device code and 
the 7-bit button code. This is achieved by a series 
of rotations then masking.  
 
The assembler version of this is shown in listing 5. 
Within the variable PACKET, the button code is 
located, starting at bit-0. This is now extracted by 
ANDing PACKET with 127 (01111111) and the 
result is placed into IR_BUT. To extract the device 
code, seven right rotations are performed, which 
will effectively move the button code out of the 
way  and place the device code starting at bit-0 of 
PACKET. Again, this is extracted by ANDing, but 
this time with 31 (00011111) and placed into 
IR_DEV. 
 
The PicBASIC Pro’s version of the same routine 
takes only two lines of code: - 
 
' Split the 7-bit Button code and the 5-bit DEVICE code 
IR_But=Packet & %01111111  ' Mask the 7 BUTTON bits 
IR_Dev=(Packet >>7) & %00011111 ' Move down and mask, the 5 DEVICE bits 
 
Task 5. 
Our finished decoder could simply bring the eight 
PortB pins high for a given button pressed on the 
handset, but a more desirable result would be to 
transmit, both the button and the device codes 
serially. Therefore, our fifth task is a subroutine 
that does just that.  
 
Listing 7 shows the assembler version of an async 
RS232 transmitter, operating at inverted 9600 baud 
from bit-3 of PortA. The byte to transmit is first 
loaded into the W register then a call is made to 
SOUT. As it stands, the baud rate is set at 9600, 
however, to change it, alter the value placed into 
DLCTR, the higher the value, the lower the baud 
rate. For instance, a value of 44 will lower it to 
4800 baud, while 88 will produce 1200 baud. 
To do the same task in PicBASIC Pro, again takes 
only one command: - 
 
Serout PortA.3 , N9600 , [ Variable ]  
 
PicBASIC Pro’s various serial out commands have 
a lot more tricks up their sleeves. Not only do they  
allow different baud rates from 600 to 19200; both inverted and non-inverted, but also output the results 
as 8 or 16-bit decimal, hexadecimal, binary or ASCII strings. This is ideal for interfacing to the many se-
rially controlled LCD modules on the market. 

; Split the 7-bit BUTTON code, and the 5-bit DEVICE code 
Movf Packet,W ; Mask the 7-bit BUTTON code 

 Andlw B’01111111’ 
 Movwf  IR_But 
; ** Shift PACKET and PACKET+1, right, 7 times ** 
 Rrf Packet+1,F 
 Rrf Packet,F 
 Rrf Packet+1,F 
 Rrf Packet,F 
 Rrf Packet+1,F 
 Rrf Packet,F 
 Rrf Packet+1,F 
 Rrf Packet,F 
 Rrf Packet+1,F 
 Rrf Packet,F 
 Rrf Packet+1,F 
 Rrf Packet,F 
 Rrf Packet+1,F 
 Rrf Packet,F 
 Movf Packet,W ; Mask the 5-bit DEVICE code 
 Andlw B’00011111’ 
 Movwf  IR_Dev 
 Return 
 

Listing 5. Assembler, Device code splitter. 

Task 6. 
Our final task is to write the main program loop which will; call the decoder subroutine, serially transmit 
both codes, and illuminate the correct LED for a chosen button pressed on the handset.  
 
An assembler version of this is shown in listing 
8. Within the loop, the returning values from 
IRIN are examined, if IR_DEV returns holding 
255 then an invalid header was detected so the 
process is repeated. If a valid header WAS de-
tected, then both IR_DEV and IR_BUT are 
transmitted using the SOUT subroutine. A 
check is than made of IR_DEV, if it’s not hold-
ing a value of one, then it is not a television re-
mote handset, and again, the process is re-
peated. If, however, the device code is for a 
television, IR_BUT is  examined, if it holds a 
value of 16 (channel-up) then the green LED is 
turned on, and the red LED is turned on if it’s 
holding 17 (channel-down). 
 
The PicBASIC Pro version is shown in listing 
9. It has exactly the same function as previously 
described. 
 
Using the subroutine, IRIN. 
Both versions of the IRIN subroutine may eas-
ily be incorporated into your own programs. A 
brief outline of the returned variables are: - 
 
CALL or GOSUB IRIN 
IR_DEV returns holding the DEVICE code 
(0..31) 
IR_BUT returns holding the BUTTON code 
(0..127) 
Both IR_DEV and IR_BUT return holding 255 if a valid header was not received. 
 

Conclusion. 
I hope that I’ve succeeded in illustrating 
that both, infrared decoding and 
PICmicro programming need not be the 
exclusive property of the whizz kids or 
rocket scientists among us. Assembly 
language will never be fully replaced by 
high level languages, especially if com-
pact or critically timed code is required. 
But with the advent of ever increasing 
speeds and memory storage on the new 
PIC ranges being developed, this is fast 
becoming a non-issue. The one major 

advantage that assembler has, is that it’s free. All the tools required for software development are 
downloadable from microchip’s web site, there are also a plethora of datasheets and application notes, 

' ** THE MAIN PROGRAM LOOP STARTS HERE ** 
Again: Low Green_LED:Low Red_LED ' Extinguish both LED's 
 Gosub IRIN   ' Receive an IR signal 
 If IR_Dev=255 then goto Again ' Check for valid header 
 If IR_Dev<>0 then goto Again ' If not a TV DEVICE code  

‘ then look again 
Serout PortA.3,N9600,[IR_Dev,IR_But]' Transmit the 2 bytes 
If IR_But=116 then High Green_LED' If channel up, then green 

‘ LED on 
 If IR_But=117 then High Red_LED ' If channel down, then red 

‘ LED on 
 Pause 10   ' Delay for 10ms (optional) 
 Goto Again   ' Do it forever 
 

Listing 9. PicBasic Pro, Main body code. 

; ** THE MAIN PROGRAM LOOP STARTS HERE ** 
Again Clrwdt  ; Walk the dog 
 Call  IRIN ; Get the IR signal from the handset 
 Bcf PortA,0 ; Turn off both LEDs 
 Bcf PortA,1 
 Movlw 255 ; If IR_DEV=255 then look again 
 Subwf IR_Dev,w 
 Btfsc STATUS,z 
 Goto Again 
; ** Transmit the DEVICE code then the BUTTON code serially 
; ** at inverted 9600 baud N-8-1 ** 
 Movf IR_Dev,w  
 Call  Sout 
 Movf IR_But,w  
 Call  Sout 
; ** If IR_DEV<>1 (TV device code) then look Again ** 
 Movlw 0 
 Subwf IR_Dev,w 
 Btfss STATUS,z 
 Goto Again 
; ** If IR_But=116 (channel up) then illuminate the green LED 
 Movlw 16 
 Subwf IR_But,w 
 Btfss STATUS,z 
 Goto CH_UP  
 Bsf PortA,1 
; **If IR_BUT=117 (channel down) then illuminate the red LED 
CH_UP Movlw 17 
 Subwf IR_But,w 
 Btfss STATUS,z 
 Goto Exit 
 Bsf PortA,0 
Exit Call Delay ; Delay for 10ms (optional) 
 Goto Again 
 

Listing 8. Assembler, Main code loop. 

which are downloadable from the same site. But this doesn’t detract from the fact that assembly language 
is somewhat difficult to learn and sometimes tedious to write. 
 
Using a high level language such as PicBASIC or PicBASIC Pro, not only makes programming a more 
enjoyable experience, but opens up a whole new aspect of electronics that was previously beyond the 
scope of all but the most advanced hobbyist, such as I2C, SPI serial eeprom, Analogue to Digital, Digital 
to Analogue interfacing among many others, the list is as long as your imagination and creativity allows. 
However, it’s not just the hobbyist who can benefit from this remarkable language. Because, both assem-
bler and BASIC may be freely mixed within the same program, extremely powerfull and flexible pro-
grams may be written that can greatly decrease prototyping time, thus reducing the overall costs of a 
commercial product. After all, time is a precious commodity that should not be wasted 
 
Resources. 
Further information on infrared encoding and decoding, including a compatible Sony infrared transmitter 
project may be found in the book, EXPERIMENTING with the PICBASIC PRO COMPILER, writ-
ten by “yours truly”. This is available from Crownhill Associates at www.crownhill.co.uk. Furthermore, 
all the components used in this project are also available from Crownhill, including the infrared sensor 
and a suitable remote handset, along with the full source code for both language versions of the infrared 
decoder. 
 
Les Johnson. 

